Problem 1

```
> restart:
> infolevel[solve]:=3:
> sols:=solve(x^3-3*x^2+3=0,x);
Main: Entering solver with 1 equation in 1 variable
Dispatch: handling polynomials of the form a*x^n-b
Dispatch: handling a single polynomial
Main: solving successful - now forming solutions
Main: Exiting solver returning 1 solution
```

$$sols := \frac{\left(-4+4\,I\sqrt{3}\right)^{1/3}}{2} + \frac{2}{\left(-4+4\,I\sqrt{3}\right)^{1/3}} + 1,\ -\frac{\left(-4+4\,I\sqrt{3}\right)^{1/3}}{4} - \frac{1}{\left(-4+4\,I\sqrt{3}\right)^{1/3}} + 1 + \frac{I\sqrt{3}\left(\frac{\left(-4+4\,I\sqrt{3}\right)^{1/3}}{2} - \frac{2}{\left(-4+4\,I\sqrt{3}\right)^{1/3}}\right)}{2},\ -\frac{\left(-4+4\,I\sqrt{3}\right)^{1/3}}{4} - \frac{1}{\left(-4+4\,I\sqrt{3}\right)^{1/3}} + 1 - \frac{I\sqrt{3}\left(\frac{\left(-4+4\,I\sqrt{3}\right)^{1/3}}{2} - \frac{2}{\left(-4+4\,I\sqrt{3}\right)^{1/3}}\right)}{2} \tag{1}$$

```
> sols:=evalc([sols]); # Use complex numbers tricks to simplify
```

$$sols := \left[2\cos\left(\frac{2\pi}{9}\right)+1,\ -\cos\left(\frac{2\pi}{9}\right)+1-\sqrt{3}\sin\left(\frac{2\pi}{9}\right),\ -\cos\left(\frac{2\pi}{9}\right)+1+\sqrt{3}\sin\left(\frac{2\pi}{9}\right)\right] \tag{2}$$

```
> evalf(sols); # Evaluate using variable-precision floating-point numbers
```

$$[2.532088886,\ -0.8793852421,\ 1.347296356] \tag{3}$$

```
> evalhf(sols); # Evaluate using 64-bit floating-point numbers (didn't work)
Error, unable to evaluate expression to hardware floats: [2*cos((2/9)*Pi)+1, -cos((2/9)*Pi)+1-3^(1/2)*sin((2/9)*Pi), -cos((2/9)*Pi)+1+3^(1/2)*sin((2/9)*Pi)]
> #infolevel[fsolve]:=5: # Doesn't print info
> fsolve(x^3-3*x^2+3=0,x);
```

$$-0.8793852416,\ 1.347296355,\ 2.532088886 \tag{4}$$

```
> plot(x^3-3*x^2+3,x=-1..3);
```



Problem 2

```
> plot(x^5-3*x^2+1,x=-2..2);
```




```
> Digits:=20: # Variable-precision arithmetic
> fsolve(x^5-3*x^2+1=0,x);
```

$$-0.56107000717028161263,\ 0.59924102796568577923,\ 1.3480469412913384769 \quad (5)$$

Why symbolic algebra is not to be mixed with floating-point computations without care

Apply secant method to Problem 1

```
> x_kp1:= (x_k,x_km1)->x_k-(x_k-x_km1)/(f(x_k)-f(x_km1))*f(x_k);
```

$$x\_kp1 := (x\_k, x\_km1) \mapsto x\_k - \frac{(x\_k - x\_km1) \cdot f(x\_k)}{f(x\_k) - f(x\_km1)} \quad (6)$$

```
> f:=x->x^3-3*x^2+3;
```

$$f := x \mapsto x^3 - 3 \cdot x^2 + 3 \quad (7)$$

```
> secant_p1:=simplify(x_kp1(x_k,x_km1));
```

$$secant\_p1 := \frac{-3 + x\_k\, x\_km1^2 + \left(x\_k^2 - 3\, x\_k\right) x\_km1}{x\_km1^2 + (x\_k - 3)\, x\_km1 + x\_k^2 - 3\, x\_k} \quad (8)$$

```
> x[0]:=1; x[1]:=2;
```

$$x_0 := 1$$

$$x_1 := 2 \tag{9}$$

The results we are computing are rational numbers, not floating-point numbers! By k=10 they are too big to print on the page!

```
> for k from 1 to 6 do x[k+1]:=x_kp1(x[k],x[k-1]); od;
```

$$x_2 := \frac{3}{2}$$

$$x_3 := \frac{6}{5}$$

$$x_4 := \frac{118}{87}$$

$$x_5 := \frac{704319}{522596}$$

$$x_6 := \frac{7319710804447662}{5432894780288459}$$

$$x_7 := \frac{954910537526246873969955803591300886028009 0}{708760573391667125860242116086496898152499 3} \tag{10}$$

```
> Digits:=20:
> evalf(x[7]);
```

$$1.3472963556037917232 \tag{11}$$

```
> fsolve(x^3-3*x^2+3,x);
```

$$-0.87938524157181676811,\ 1.3472963553338606977,\ 2.5320888862379560704 \tag{12}$$

Problem 3

```
> restart:
> x_kp1:= (x_k,x_km1)->x_k-(x_k-x_km1)/(f(x_k)-f(x_km1))*f(x_k);
```

$$x\_kp1 := (x\_k, x\_km1) \mapsto x\_k - \frac{(x\_k - x\_km1) \cdot f(x\_k)}{f(x\_k) - f(x\_km1)} \tag{13}$$

Part b

```
> psi:=(x_k,x_km1)->(x_kp1(x_k,x_km1)-xi)/(x_k-xi)/(x_km1-xi);
```

$$\psi := (x\_k, x\_km1) \mapsto \frac{x\_kp1(x\_k, x\_km1) - \xi}{(x\_k - \xi) \cdot (x\_km1 - \xi)} \tag{14}$$

```
> psi_expr:=simplify(psi(x_k,x_km1));
```

$$psi\_expr := \frac{x\_k - \frac{(x\_k - x\_km1)\, f(x\_k)}{f(x\_k) - f(x\_km1)} - \xi}{(x\_k - \xi)\ (x\_km1 - \xi)} \tag{15}$$

To complete part b we need to use that f(xi)=0

```
> limit(psi_expr,x_k=xi); # Maple cannot do calculation!
```

$$\lim_{x\_k \to \xi} \frac{x\_k - \frac{(x\_k - x\_km1)\, f(x\_k)}{f(x\_k) - f(x\_km1)} - \xi}{(x\_k - \xi)\ (x\_km1 - \xi)} \tag{16}$$

```
> psi_series:=convert(series(psi_expr, x_k=xi, 2),polynom);
```

(17)

$$psi\_series := -\frac{(-x\_km1+\xi)\, f(\xi)}{(f(\xi)-f(x\_km1))\,(x\_km1-\xi)\,(x\_k-\xi)} + \frac{1-\frac{(-x\_km1+\xi)\, \mathrm{D}(f)(\xi)}{f(\xi)-f(x\_km1)} - \frac{\left(1+\frac{(x\_km1-\xi)\,\mathrm{D}(f)(\xi)}{f(\xi)-f(x\_km1)}\right) f(\xi)}{f(\xi)-f(x\_km1)}}{x\_km1-\xi} \tag{17}$$

```
> infolevel[simplify]:=0:
> psi_limit_Maple:=simplify(psi_series,{f(xi)=0}); # Now use that f
  (xi)=0 to help Maple
```

$$psi\_limit\_Maple := \frac{(x\_km1-\xi)\,\mathrm{D}(f)(\xi)-f(x\_km1)}{(-x\_km1+\xi)\, f(x\_km1)} \tag{18}$$

```
> eval(psi_series, f(xi)=0);
```

$$\frac{1+\frac{(-x\_km1+\xi)\,\mathrm{D}(f)(\xi)}{f(x\_km1)}}{x\_km1-\xi} \tag{19}$$

Now let's try to do this ourselves in a smart way

```
> psi_expr;
```

$$\frac{x\_k-\frac{(x\_k-x\_km1)\, f(x\_k)}{f(x\_k)-f(x\_km1)}-\xi}{(x\_k-\xi)\,(x\_km1-\xi)} \tag{20}$$

```
> piece_1:=eval(psi_expr, f(x_k)=0);
```

$$piece\_1 := \frac{1}{x\_km1-\xi} \tag{21}$$

```
> piece_2:=simplify(psi_expr-piece_1);
```

$$piece\_2 := -\frac{(x\_k-x\_km1)\, f(x\_k)}{(f(x\_k)-f(x\_km1))\,(x\_k-\xi)\,(x\_km1-\xi)} \tag{22}$$

```
> problem:=f(x_k)/(x_k - xi);
```

$$problem := \frac{f(x\_k)}{x\_k-\xi} \tag{23}$$

```
> piece_3:=simplify(piece_2/problem);
```

$$piece\_3 := \frac{x\_k-x\_km1}{(f(x\_k)-f(x\_km1))\,(-x\_km1+\xi)} \tag{24}$$

```
> psi_new:=piece_1+problem*piece_3;
```

$$psi\_new := \frac{(x\_k-x\_km1)\, f(x\_k)}{(f(x\_k)-f(x\_km1))\,(-x\_km1+\xi)\,(x\_k-\xi)} + \frac{1}{x\_km1-\xi} \tag{25}$$

```
> simplify(psi_expr-psi_new); # They are the same
```

$$0 \tag{26}$$

```
> simplify(series(problem, x_k=xi, 2), {f(xi)=0});
```

$$\mathrm{D}(f)(\xi)+\mathrm{O}((x\_k-\xi)) \tag{27}$$

```
> psi_limit:=simplify(eval(piece_1+D(f)(xi)*piece_3, x_k=xi));
```

$$psi\_limit := \frac{1}{x\_km1-\xi}+\frac{\mathrm{D}(f)(\xi)}{f(\xi)-f(x\_km1)} \tag{28}$$

Part c

```
> limit(psi_limit, x_km1=xi);
```

$$\frac{\mathrm{D}^{(2)}(f)(\xi)}{2\,\mathrm{D}(f)(\xi)} \quad (29)$$

Part d

```
> x_k_m_xi_kp1:=k->A*(x[k-1]-xi)^q;
```

$$x\_k\_m\_xi\_kp1 := k \mapsto A \cdot \left(x_{k-1} - \xi\right)^q \quad (30)$$

```
> psi_limit_assumpt:=x_k_m_xi_kp1(k+1)/x_k_m_xi_kp1(k)/(x[k-1]-xi);
```

$$psi\_limit\_assumpt := \frac{\left(x_k - \xi\right)^q}{\left(x_{k-1} - \xi\right)^q \left(x_{k-1} - \xi\right)} \quad (31)$$

```
> psi_limit_assumpt := x_k_m_xi_kp1(k)^q/((x[k - 1] - xi)^q*(x[k -
  1] - xi));
```

$$psi\_limit\_assumpt := \frac{\left(A \left(x_{k-1} - \xi\right)^q\right)^q}{\left(x_{k-1} - \xi\right)^q \left(x_{k-1} - \xi\right)} \quad (32)$$

```
> simplify(psi_limit_assumpt, symbolic);
```

$$A^q \left(x_{k-1} - \xi\right)^{q^2 - q - 1} \quad (33)$$

```
> solve(q^2 - q - 1, q);
```

$$\frac{\sqrt{5}}{2} + \frac{1}{2},\ -\frac{\sqrt{5}}{2} + \frac{1}{2} \quad (34)$$

Importance of assumptions in symbolic algebra:

```
> simplify(sqrt(q^2)) assuming q>0;
```

$$q \quad (35)$$

```
> simplify(sqrt(q^2),symbolic);
```

$$q \quad (36)$$